Properties of finite amplitude electromagnetic waves propagating in the quantum vacuum

https://doi.org/10.1088/1361-6587/ab21fb
Notebook: Óscar Amaro, June 2021, November 2022 @ GoLP-EPP
Contact: oscar.amaro@tecnico.ulisboa.pt

Introduction

In this notebook we reproduce some results from the paper.

Prove eq 16+17 is equivalent to eq 13+14 ...

```
(* eq 16 is the easiest to prove *)
Clear[𝔼, 𝔹, E0, B0, Ez, By, a, b, x, t, α, β, γ]

(* eq 15 *)
Ez[x, t] = E0 + a[x, t];
By[x, t] = B0 + b[x, t];

(* eq 13 *)
D[By[x, t], t] - D[Ez[x, t], x]
```

Out[147]= $b^{(0,1)}[x, t] - a^{(1,0)}[x, t]$

In[1538]:=
```
(* eq 17 requires a bit more work ... *)
Clear[𝔼, 𝔹, E0, B0, Ez, By, a, b, x, t, α, β, γ, eq14, eq17, κ1, κ2]
κ2 = 180/315 κ1;

(* eq 15 *)
Ez[x, t] = E0 + a[x, t];
By[x, t] = B0 + b[x, t];

(* eq 14 *)
eq14 = - (1 + 8 κ1 Ez[x, t]^2 + 4 (Ez[x, t]^2 - By[x, t]^2) (κ1 - 3 κ2 Ez[x, t]^2) -
      6 κ2 (Ez[x, t]^2 - By[x, t]^2)^2) D[Ez[x, t], t] +
    (1 - 8 κ1 By[x, t]^2 + 4 (Ez[x, t]^2 - By[x, t]^2) (κ1 + 3 κ2 By[x, t]^2) -
      6 κ2 (Ez[x, t]^2 - By[x, t]^2)^2) D[By[x, t], x] +
    4 × (2 κ1 - 3 κ2 (Ez[x, t]^2 - By[x, t]^2)) Ez[x, t] × By[x, t] ×
     (D[By[x, t], t] + D[Ez[x, t], x]) // Expand // Simplify;

(* eq 18 *)
α = 1 + 8 κ1 (E0 + a[x, t])^2 + 4 ((E0 + a[x, t])^2 - (B0 + b[x, t])^2)
    (κ1 - 3 (E0 + a[x, t])^2 κ2) - 6 κ2 ((E0 + a[x, t])^2 - (B0 + b[x, t])^2)^2;
(* eq 19 *)
β = 4 (E0 + a[x, t]) (B0 + b[x, t]) (2 κ1 - 3 κ2 ((E0 + a[x, t])^2 - (B0 + b[x, t])^2));
(* eq 20 *)
γ = 1 - 8 κ1 (B0 + b[x, t])^2 + 4 ((E0 + a[x, t])^2 - (B0 + b[x, t])^2)
    (κ1 + 3 (B0 + b[x, t])^2 κ2) - 6 κ2 ((E0 + a[x, t])^2 - (B0 + b[x, t])^2);

(* eq 17 *)
eq17 = α D[a[x, t], t] - β (D[a[x, t], x] + D[b[x, t], t]) - γ D[b[x, t], x] // Expand //
   Simplify;

(* confirm eq 21 *)
{α, β, γ} /. {a[x, t] → 0, b[x, t] → 0, B0 → E0}

res = eq14 / eq17;
(res /. {b^(0,1)[x, t] → a^(1,0)[x, t]}) // FullSimplify

(*
E0=RandomReal[];
B0=RandomReal[];
κ1=RandomReal[];
res//FullSimplify
*)
```

Out[1547]= $\{1 + 8\, E0^2\, \kappa 1,\ 8\, E0^2\, \kappa 1,\ 1 - 8\, E0^2\, \kappa 1\}$

Out[1549]=
$$\Big(\Big(7-4\times\left(7+6\,B0^2-6\,E0^2\right)\left(B0^2-3\,E0^2\right)\kappa1+4\,\kappa1\,\Big(-72\,E0\,a[x,t]^3-18\,a[x,t]^4-b[x,t]\,(2\,B0+b[x,t])\times\left(7+12\,B0^2-24\,E0^2+6\,b[x,t]\,(2\,B0+b[x,t])\right)+3\,a[x,t]^2\left(7+8\,B0^2-36\,E0^2+8\,b[x,t]\,(2\,B0+b[x,t])\right)+6\,E0\,a[x,t]\left(7+8\,B0^2-12\,E0^2+8\,b[x,t]\,(2\,B0+b[x,t])\right)\Big)\Big)\,a^{(0,1)}[x,t]-7\,b^{(1,0)}[x,t]+4\,\kappa1\left(7+6\,B0^2-6\,E0^2-6\,a[x,t]\,(2\,E0+a[x,t])+6\,b[x,t]\,(2\,B0+b[x,t])\right)\times\Big(-4\,(E0+a[x,t])\,(B0+b[x,t])\,a^{(1,0)}[x,t]-\left(-3\,B0^2+E0^2+2\,E0\,a[x,t]+a[x,t]^2-3\,b[x,t]\,(2\,B0+b[x,t])\right)b^{(1,0)}[x,t]\Big)\Big)\Big/$$
$$\Big(\Big(-7+4\times\left(7+6\,B0^2-6\,E0^2\right)\left(B0^2-3\,E0^2\right)\kappa1+4\,\kappa1\,\Big(72\,E0\,a[x,t]^3+18\,a[x,t]^4+b[x,t]\,(2\,B0+b[x,t])\times\left(7+12\,B0^2-24\,E0^2+6\,b[x,t]\,(2\,B0+b[x,t])\right)-3\,a[x,t]^2\left(7+8\,B0^2-36\,E0^2+8\,b[x,t]\,(2\,B0+b[x,t])\right)-6\,E0\,a[x,t]\left(7+8\,B0^2-12\,E0^2+8\,b[x,t]\,(2\,B0+b[x,t])\right)\Big)\Big)\,a^{(0,1)}[x,t]+16\,\kappa1\,(E0+a[x,t])\,(B0+b[x,t])\left(7+6\,B0^2-6\,E0^2-6\,a[x,t]\,(2\,E0+a[x,t])+6\,b[x,t]\,(2\,B0+b[x,t])\right)a^{(1,0)}[x,t]+\Big(7+4\left(E0^2-3\,B0^2\left(5+4\,B0^2-4\,E0^2\right)\right)\kappa1+4\,\kappa1\,\Big(-3\,b[x,t]\,(2\,B0+b[x,t])\times\left(5+8\,B0^2-4\,E0^2+4\,b[x,t]\,(2\,B0+b[x,t])\right)+2\,E0\,a[x,t]\left(1+12\,B0^2+12\,b[x,t]\,(2\,B0+b[x,t])\right)+a[x,t]^2\left(1+12\,B0^2+12\,b[x,t]\,(2\,B0+b[x,t])\right)\Big)\Big)\,b^{(1,0)}[x,t]\Big)$$

In[1319]:=
```
(* get eq 23 *)
Clear[eq16, eq17, x, t, a, b, α, β, γ, aa, bb, ν, q, ω, α0, β0, γ0]
(* eq 22 ansatz *)
a[x, t] = aa Exp[-I ω t + I q x];
b[x, t] = bb Exp[-I ω t + I q x];

eq16 = D[b[x, t], t] - D[a[x, t], x] // Simplify;
eq17 =
  α D[a[x, t], t] - β (D[a[x, t], x] + D[b[x, t], t]) - γ D[b[x, t], x] // Expand //
   Simplify;
ω = ν q;
(* *)
```
$$\frac{eq16}{aa+\nu\,bb}\ \text{// Simplify}$$
$$\frac{eq17}{\nu\,(bb\,\beta-aa\,\alpha)-(aa\,\beta+bb\,\gamma)}\ \text{// FullSimplify}$$

Out[1325]= $-\mathrm{i}\,e^{\mathrm{i}\,q\,(x-t\,\nu)}\,q$

Out[1326]= $\mathrm{i}\,e^{\mathrm{i}\,q\,(x-t\,\nu)}\,q$

In[774]:=
```
(* Taylor expand eq 69 ... *)
Clear[κ1, E0, E02]
Series[(1 - 8 κ1 E0^2)/(1 + 8 κ1 E0^2), { E0, 0, 4}]
(1 - 8 κ1 E02) × (1 - 8 κ1 E02) // Expand
```

Out[775]= $1-16\,\kappa 1\,\mathrm{E0}^2+128\,\kappa 1^2\,\mathrm{E0}^4+O[\mathrm{E0}]^5$

Out[776]= $1-16\,\mathrm{E02}\,\kappa 1+64\,\mathrm{E02}^2\,\kappa 1^2$

In[1180]:=
```
(* get eq 24 *)
Clear[a, b, ν, α0, β0, γ0, sol]
a = -ν b;
sol = Solve[ν (b β0 - a α0) - (a β0 + b γ0) == 0, ν]

(* use relationships by equation 21 *)
{α0, β0, γ0} = {1 + 8 E0^2 κ1, 8 E0^2 κ1, 1 - 8 E0^2 κ1};
(* get eq 25 *)
sol // Simplify
```

Out[1182]= $\left\{\left\{\nu\to\frac{-\beta 0-\sqrt{\beta 0^2+\alpha 0\,\gamma 0}}{\alpha 0}\right\},\left\{\nu\to\frac{-\beta 0+\sqrt{\beta 0^2+\alpha 0\,\gamma 0}}{\alpha 0}\right\}\right\}$

Out[1184]= $\left\{\{\nu\to -1\},\left\{\nu\to\frac{1-8\,\mathrm{E0}^2\,\kappa 1}{1+8\,\mathrm{E0}^2\,\kappa 1}\right\}\right\}$

```
In[883]:= (* get eq 71 ... *)
(*
Clear[κ1,κ2,E0,ν,fp]
 ν=(1-8κ1 E0^2)/(1+8 κ1 E0^2);
fp=4E0(κ1(-1-3ν-3ν^2+3/ν)+3κ2 E0^2(ν^2+3ν-3+1/ν))//Simplify;
*)

Clear[κ1, κ2, E0, ν, fp]
(* the correct coefficient of E0^4 will not change eq 71 *)
ν = 1 - 16 κ1 E0^2 + 128 κ1^2 E0^4;
fp = 4 E0 (κ1 (-1 - 3 ν - 3 ν^2 + 3/ν) + 3 κ2 E0^2 (ν^2 + 3 ν - 3 + 1/ν)) // Expand;

(* f' will not be a polynome in E0 because of the rational form of ν *)

(* this is not equal to eq 71 ...*)
CoefficientList[
  Normal[Series[fp, {E0, 0, 4}]] - 48 E0^3 (12 κ1^2 + κ2) // Expand, E0][[4]]

(* but 48 E0^3(16κ1^2+0.5κ2) is *)
Normal[Series[fp, {E0, 0, 4}]] - 48 E0^3 (16 κ1^2 + 0.5 κ2) // Expand
Out[886]= 192 κ1^2 - 24 κ2
Out[887]= 0. - 16 E0 κ1
```

Figure 1: visualize eq 73

In[1148]:=
```
Clear[a0, am, fp, x, t]

a0 = am Sin[x0];
x = x0 + (1 + fp a0) t

(* a(x,x0=0) is the first (to the left) point of each curve. for Δt=
 0.5 (difference in time is uniform for all curves) then a(x,x0=0) progresses
   linearly. the first blue curve for lowest time is for t=0.5 and not t=0 *)
x /. {x0 → 0}

(* looking at the plot, the amplitude am cannot be 1,
but needs to be around 1.5, which is very close to π/2 *)
am = π / 2;
fp = -0.35;

tab = Table[{x, a0 // N}, {x0, 0, 2 π, 2 π / 100}];
ListPlot[{(tab /. {t → 0.5}), (tab /. {t → 1}), (tab /. {t → 1.5}), (tab /. {t → 2}),
   (tab /. {t → 2.5}), (tab /. {t → 3}), (tab /. {t → 3.5}), (tab /. {t → 0})},
 Joined → True, Frame → True, FrameLabel → {"x", "a"},
 GridLines → Automatic, AspectRatio → 0.3, ImageSize → Large]
```

Out[1150]= x0 + t (1 + am fp Sin[x0])

Out[1151]= t

Out[1155]=



Figure 3

In[448]:=
```
(* eq 87 *)
Clear[a1, a10, t, fp]
DSolve[{D[a1[t], t] == -fp a1[t]^2, a1[0] == a10}, a1[t], t]
```

Out[449]= $\left\{\left\{\text{a1}[t] \rightarrow \frac{\text{a10}}{1 + \text{a10 fp } t}\right\}\right\}$

```
In[32]:= Clear[a0, a1, a10, fp, tab]
a1[t_, a10_, fp_] := a10 / (1 + fp a10 t)
a[x_, t_, a0_, a10_, fp_] := a0 + a1[t, a10, fp] x

(* a) *)
tab = Table[a[x, t, -3 t + 1.8, 1, -0.5], {t, 0.2, 1.2, 0.1}];
Plot[tab, {x, 0, 10}, Frame → True, FrameLabel → {"x", "a"},
 PlotLabel → "(a) The rarefaction waves breaks."]

(* b) *)
tab = Table[a[x, t, -3 t + 1.8, -1, 0.5], {t, 0.2, 1.2, 0.1}];
Plot[tab, {x, 0, 10}, Frame → True, FrameLabel → {"x", "a"},
 PlotLabel → "(b) The rarefaction waves do not break."]
```

Out[36]=



Out[38]=



Figure 4

In[429]:=
```
(* σγγep in "natural" units, and assuming ω=Ω or ω1=ω2 *)
Clear[σγγep, re, e, me, c, ħ, ω, βe]
βe = Sqrt[1 - me^2 c^4/(ħ^2 ω^2)];
σγγep = π re^2/2 (1 - βe^2) × ((3 - βe^4) Log[(1 + βe)/(1 - βe)] - 2 βe (2 - βe^2));
re = e^2 / (me c^2);
e = me = c = ħ = 1;
Plot[σγγep, {ω, 0, 10}, PlotRange → {0, 2.5},
 Frame → True, FrameLabel → {"ħω/(me c^2)", "σγγep"}]

Plot[{σγγep, π re^2 Sqrt[ħ^2 ω^2/(me^2 c^4) - 1]}, {ω, 0, 2}, PlotRange → {{0, 2}, {0, 5}},
 Frame → True, FrameLabel → {"ħω/(me c^2)", "σγγep"}, ImageSize → Small,
 PlotLegends → {"Exact cross section: eq 90", "Near-threshold: eq 89"},
 PlotStyle → {Automatic, Red}]
```

Out[434]=



Out[435]=